;     Функционално програмиране.

;     Упражнение N9

;     Дефиниции на функции за работа със списъци на произволно

;ниво на вложение.

;Задача 1. Да се дефинира функция, която действа като вградената

;функция length.

(define (user-length l)

  (if (null? l)

      0

      (apply + (map (lambda (x) 1) l))))

;Задача 2. Да се дефинира функция, която намира броя на атомите в

;даден списък, намиращи се в списъка на произволно ниво на

;вложение.
(require (lib "compat.ss"))

(define (count-atoms l)

  (cond ((null? l) 0)

        ((atom? l) 1)

        (else (apply + (map count-atoms l))) ))

;Задача 3. Да се дефинира функция, която намира максималната

;дълбочина на вложение в даден списък.

(define (depth l)

  (cond ((null? l) 1)

        ((atom? l) 0)

        (else (+ 1 (apply max (map depth l)))) ))

;Задача 4. Да се дефинира функция, която има два аргумента: списък

;от списъци ll и s-израз s, и като резултат добавя s-израза s

;отпред към всеки от списъците - елементи на списъка ll.

;Първи вариант:

(define (scons s ll)

  (if (null? ll)

      ()
      (cons (cons s (car ll)) (scons s (cdr ll))) ))

;Втори вариант:

(define (scons1 s ll)

  (map (lambda (l) (cons s l)) ll))

;Задача 5. Даден е списък, на който елементите са списъци от числа.

;Да се намери най-малкия елемент от списъка, образуван от най-

;големите елементи на всеки подсписък.

(define (minmax l)

  (apply min (map (lambda (x) (apply max x)) l)))

;Задача 6. Даден е списък, на който елементите са числа и списъци

;от числа. Да се заместят списъците със сумите от елементите им.

;Например от списъка (2 (3 4) (1 1) 5) се получава списъкът 

;(2 7 2 5).

(define (f x)

  (if (atom? x)

      x

      (apply + x)))

(define (vect l)

  (map f l))

;Задача 7. Да се напише функция от по-висок ред scan, която по

;дадени бинарна операция op, параметър а и редица (х1 х2 ... хn),

;връща нова редица, получена чрез прилагане на правилото

;(scan op a (x1 x2 ... xn)  =  (a  a op x1  (a op x1) op x2  ...  )

(define (scan op a l)

  (if (null? l)

      (list a)

      (cons a (scan op (op a (car l)) (cdr l)))))

;Задача 8. Като се използва предишната задача, да се дефинира

;функция inits, която се прилага по следното правило:

;(inits (x1 x2 ... xn)) -> ( () (x1) (x1 x2) ... )

(define (inits l)

  (scan insertlast '() l))

(define (insertlast l elem)

  (append l (list elem)))

;Задача 9. Като се използва предишната задача, да се дефинира

;функция tails, която се прилага по правилото:

;(tails (x1 x2 ... xn)) -> ((x1 x2 ... xn) (x2 ... xn) ... (xn) ())

(define (tails l)

  (scan (lambda (x y) (cdr x)) l l))

